#include <WiFi.h>
#include <WebServer.h>
#include <SD_MMC.h>
#include "esp_camera.h"
#include "Arduino.h"
#include "FS.h" 
#include "SD_MMC.h"
#include "soc/soc.h"
#include "soc/rtc_cntl_reg.h" 
#include "driver/rtc_io.h"
#include <EEPROM.h>
#define EEPROM_SIZE 1
 
RTC_DATA_ATTR int bootCount = 0;
int pictureNumber = 0;
 
#define PWDN_GPIO_NUM     32
#define RESET_GPIO_NUM    -1
#define XCLK_GPIO_NUM      0
#define SIOD_GPIO_NUM     26
#define SIOC_GPIO_NUM     27
#define Y9_GPIO_NUM       35
#define Y8_GPIO_NUM       34
#define Y7_GPIO_NUM       39
#define Y6_GPIO_NUM       36
#define Y5_GPIO_NUM       21
#define Y4_GPIO_NUM       19
#define Y3_GPIO_NUM       18
#define Y2_GPIO_NUM        5
#define VSYNC_GPIO_NUM    25
#define HREF_GPIO_NUM     23
#define PCLK_GPIO_NUM     22
/*
const char *AP_SSID = "ESP32-CAM";
const char *AP_PASS = "12345678";
IPAddress apIP(192, 168, 4, 1); //设置AP的IP地址
*/


WebServer server(80); //声明WebServer对象
String getContentType(String path) //解析客户端请求的ContentType
{
  if (path.endsWith(".htm"))
  {
    return "text/html";
  }
  else if (path.endsWith(".html"))
  {
    return "text/html";
  }
  else if (path.endsWith(".css"))
  {
    return "text/css";
  }
  else if (path.endsWith(".js"))
  {
    return "text/javascript";
  }
  else if (path.endsWith(".png"))
  {
    return "image/png";
  }
  else if (path.endsWith(".gif"))
  {
    return "image/gif";
  }
  else if (path.endsWith(".jpg"))
  {
    return "image/jpeg";
  }
  else if (path.endsWith(".ico"))
  {
    return "image/x-icon";
  }
  else if (path.endsWith(".xml"))
  {
    return "text/xml";
  }
  else if (path.endsWith(".pdf"))
  {
    return "application/pdf";
  }
  else if (path.endsWith(".zip"))
  {
    return "application/zip";
  }
  else if (path.endsWith(".gz"))
  {
    return "application/x-gzip";
  }
  return "text/plain";
}

bool loadFromSdCard(String path) //从SD卡读取文件发送给客户端
{
  String contentType = getContentType(path);
  File contentFile = SD_MMC.open(path);
  if ((!contentFile) || contentFile.isDirectory())
  {
    Serial.print("读取文失败：");
    Serial.println(path);
    return false;
  }
  Serial.print("读取到文件：");
  Serial.println(path);
  server.streamFile(contentFile, contentType);
  contentFile.close();

  return true;
}

void handleRoot() //回调函数
{
  Serial.print("正在发送界面....");
  String message="<html><head><meta charset=\"UTF-8\"><meta name=\"viewport\" content=\"width=device-width, initial-scale=1.0, minimum-scale=0.5, maximum-scale=2.0, user-scalable=yes\" /></head><body><h2>图片：</h2>";
  File root = SD_MMC.open("/picture");
  File file = root.openNextFile();
  while (file)
  {
    message += "<img src=\"/picture/" + String(file.name()) + "\" alt=\"" + String(file.name()) + "\" width=\"200\" height=\"200\"/>";
    file.close();
    file = root.openNextFile();
    Serial.print("读取中...");
  }
  message += "</body></html>";
  server.send(200,"text/html",message);
  file.close();
  root.close();
  Serial.print("发送完毕！");
}

void handleNotFound() //未知链接回调函数，在这里相当于所有html文件相关联的内容都通过这里访问
{
  if (loadFromSdCard(server.uri()))
  {
    return;
  }
  server.send(404, "text/plain", "访问的内容不存在");
}

void setup()
{
  WRITE_PERI_REG(RTC_CNTL_BROWN_OUT_REG, 0); //disable brownout detector
  Serial.begin(115200);
  camera_config_t config;
  config.ledc_channel = LEDC_CHANNEL_0;
  config.ledc_timer = LEDC_TIMER_0;
  config.pin_d0 = Y2_GPIO_NUM;
  config.pin_d1 = Y3_GPIO_NUM;
  config.pin_d2 = Y4_GPIO_NUM;
  config.pin_d3 = Y5_GPIO_NUM;
  config.pin_d4 = Y6_GPIO_NUM;
  config.pin_d5 = Y7_GPIO_NUM;
  config.pin_d6 = Y8_GPIO_NUM;
  config.pin_d7 = Y9_GPIO_NUM;
  config.pin_xclk = XCLK_GPIO_NUM;
  config.pin_pclk = PCLK_GPIO_NUM;
  config.pin_vsync = VSYNC_GPIO_NUM;
  config.pin_href = HREF_GPIO_NUM;
  config.pin_sscb_sda = SIOD_GPIO_NUM;
  config.pin_sscb_scl = SIOC_GPIO_NUM;
  config.pin_pwdn = PWDN_GPIO_NUM;
  config.pin_reset = RESET_GPIO_NUM;
  config.xclk_freq_hz = 20000000;
  config.pixel_format = PIXFORMAT_JPEG;
  pinMode(4, INPUT_PULLDOWN);
  WRITE_PERI_REG(RTC_CNTL_BROWN_OUT_REG, 0);
  esp_sleep_enable_ext0_wakeup(GPIO_NUM_13, 1);  //外部唤醒
/*-------------------------------WIFI---------------------------------*/
  WiFi.mode(WIFI_STA);
  WiFi.begin("esp", "88888888");
  Serial.println("等待wifi连接");
  delay(3000);
  if (WiFi.status() != WL_CONNECTED){
    pinMode(4, INPUT);
    digitalWrite(4, LOW);
    rtc_gpio_hold_dis(GPIO_NUM_4);

    pinMode(13, INPUT);


    if(psramFound()){
      config.frame_size = FRAMESIZE_UXGA;
      config.jpeg_quality = 10;
      config.fb_count = 2;
    } else {
      config.frame_size = FRAMESIZE_SVGA;
      config.jpeg_quality = 12;
      config.fb_count = 1;
    }

    // Init Camera
    esp_err_t err = esp_camera_init(&config);
    if (err != ESP_OK) {
      Serial.printf("Camera init failed with error 0x%x", err);
      return;
    }

    Serial.println("Starting SD Card");

    delay(500);
    if(!SD_MMC.begin()){
      Serial.println("SD Card Mount Failed");
      //return;
    }

    uint8_t cardType = SD_MMC.cardType();
    if(cardType == CARD_NONE){
      Serial.println("No SD Card attached");
      return;
    }
    
    camera_fb_t * fb = NULL;

    fb = esp_camera_fb_get();  
    if(!fb) {
      Serial.println("Camera capture failed");
      return;
    }
    EEPROM.begin(EEPROM_SIZE);
    pictureNumber = EEPROM.read(0) + 1;

    String path = "/picture/picture" + String(pictureNumber) +".jpg";

    fs::FS &fs = SD_MMC;
    Serial.printf("Picture file name: %s\n", path.c_str());

    File file = fs.open(path.c_str(), FILE_WRITE);
    if(!file){
      Serial.println("Failed to open file in writing mode");
    }
    else {
      file.write(fb->buf, fb->len);
      Serial.printf("Saved file to path: %s\n", path.c_str());
      EEPROM.write(0, pictureNumber);
      EEPROM.commit();
    }
    file.close();
    esp_camera_fb_return(fb);
    
    delay(1000);
    
    pinMode(4, OUTPUT);
    digitalWrite(4, LOW);
    rtc_gpio_hold_en(GPIO_NUM_4);
    Serial.println("Going to sleep now");
    delay(1000);
    esp_deep_sleep_start();
    Serial.println("This will never be printed");
  }
  Serial.println("WiFi connected!");

  if (!SD_MMC.begin()) //挂载SD卡
  {
    Serial.println("存储卡挂载失败");
    return;
  }
/*-------------------------------Web---------------------------------*/
  server.on("/", handleRoot);                        //注册链接和回调函数
  server.onNotFound(handleNotFound);                 //未知链接回调函数注册

  server.begin(); //启动Web服务器
  Serial.println("Web服务器启动");

}

void loop()
{
  server.handleClient(); //处理来自客户端的请求
  if (WiFi.status() != WL_CONNECTED)
  { 
    Serial.println("检测到wifi断开, 开始深度睡眠");
    esp_deep_sleep_start();
  }
}